حل المعادلات التفاضلية:

يمكن لبرنامج المابل حل المعادلات التفاضلية الاعتيادية والجزئية كما سنرى في هذا الفصل من خلال الامثلة الاتية:

مثال(١):

تأمل المعادلة التفاضلية الاعتيادية الآتية $$\frac{1}{y^2}dy = \sin t\, dt$$

والتي يمكن حلها بطريقة فصل المتغيرات كالآتي

$$\int \frac{1}{y^2}dy = \int \sin t\, dt$$

$$-\frac{1}{y} = -\cos t + C$$

$$y = \frac{1}{\cos t + C}.$$

في برنامج المابل يمكن حل المعادلة أعلاه باستخدام الأمر **dsolve**

```
sola:=dsolve(diff(y(t),t)-y(t)^2*sin(t)=0,y(t));
```

$$sola := y(t) = (\cos(t) + \_C1)^{-1}$$

رسم الحل يكون بالأمر الآتي

```
toplota:=seq(subs(_C1=i,rhs(sola)),i=2..10);
```

هنا أعطينا الثابت C1_ أكثر من قيمة

```
> sola:=dsolve(diff(y(t),t)-y(t)^2*sin(t)=0,y(t));
toplota:=seq(subs(_C1=i,rhs(sola)),i=2..10);
plot([toplota],t=0..2*Pi,view=[0..2*Pi,0..1],color=black);
```

$$sola := y(t) = \frac{1}{\cos(t) + \_C1}$$

$$toplota := \frac{1}{\cos(t)+2}, \frac{1}{\cos(t)+3}, \frac{1}{\cos(t)+4}, \frac{1}{\cos(t)+5}, \frac{1}{\cos(t)+6}, \frac{1}{\cos(t)+7}, \frac{1}{\cos(t)+8}, \frac{1}{\cos(t)+9}, \frac{1}{\cos(t)+10}$$



مثال(٢): حل المعادلات التفاضلية الخطية من الرتبة الأولى والدرجة الأولى $\frac{d}{dt} y(t) + p(t)\, y(t) = q(t)$

```
> LODE:=diff(y(t),t)+p(t)*y(t)=q(t);
dsolve(LODE,y(t));
```

$$LODE := \frac{d}{dt} y(t) + p(t)\, y(t) = q(t)$$

$$y(t) = \left(\int q(t)\, e^{\int p(t)\, dt}\, dt + \_C1 \right) e^{\int (-p(t))\, dt}$$

يمكن حل المعادلة أعلاه مع شروط ابتدائية كما في المثال الأتي:

مثال(٣): حل المعادلة التفاضلية الأتي باستخدام برنامج المابل

$ode := \frac{d^2}{dx^2} y(x) = 2\, y(x) + 1$ مع الشروط الابتدائية $ics := y(0) = 1, D(y)(0) = 0$.

الحل:

```
> ode := diff(y(x),x,x) = 2*y(x) + 1;
ics := y(0)=1, D(y)(0)=0;
dsolve({ode,ics});
```

$$ode := \frac{d^2}{dx^2} y(x) = 2\, y(x) + 1$$

$$ics := y(0) = 1, D(y)(0) = 0$$

$$y(x) = \frac{3}{4} e^{\sqrt{2}\, x} + \frac{3}{4} e^{-\sqrt{2}\, x} - \frac{1}{2}$$

مثال (٤): استخدم تحويل لابلاس لحل المعادلة التفاضلية الآتية $\frac{d^2}{dx^2} y(x) = 2\, y(x) + 1$

```
> sol := dsolve({ode, ics}, y(x), method=laplace);
```

$$sol := y(x) = -\frac{1}{2} + \frac{3}{2}\cosh\left(\sqrt{2}\, x\right)$$

كذلك يمكن إيجاد الحل بالمتسلسلات للمثال أعلاه بالأمر التي:

```
> series_sol := dsolve({ode,ics}, y(x), series);
```

$$series\_sol := y(x) = 1 + \frac{3}{2}x^2 + \frac{1}{4}x^4 + \mathrm{O}\left(x^6\right)$$

مثال (٥): حل نظام المعادلات التفاضلية الآتية $\frac{d}{dt}y(t) = x(t),\ \frac{d}{dt}x(t) = -x(t)$

```
> sys_ode := diff(y(t),t) = x(t),  diff(x(t),t) = -x(t);
dsolve([sys_ode]);
```

$$sys\_ode := \frac{d}{dt}y(t) = x(t),\ \frac{d}{dt}x(t) = -x(t)$$

$$\left\{x(t) = \_C2\ e^{-t}, y(t) = -\_C2\ e^{-t} + \_C1\right\}$$

كذلك يمكن حل المثال أعلاه مع الشروط الابتدائية الآتية $ics := x(0) = 1, y(1) = 0$ وكالاتي:

```
> sys_ode := diff(y(t),t) = x(t),  diff(x(t),t) = -x(t);
ics := x(0)=1, y(1)=0;
dsolve([sys_ode, ics]);
```

$$sys\_ode := \frac{d}{dt}y(t) = x(t),\ \frac{d}{dt}x(t) = -x(t)$$

$$ics := x(0) = 1, y(1) = 0$$

$$\left\{x(t) = e^{-t}, y(t) = -e^{-t} + \frac{1}{e}\right\}$$

مثال (٦): حل المعادلة التفاضلة الآتية $(\cos x + 2xe^y)\,dx + \left(\sin y + x^2e^y - 1\right)dy = 0$

المعادلة أعلاه هي معادلة تفاضلية تامة بالشكل $M(x,y)dx + N(x,y)dy = 0$

لأنها تحقق الشرط $\partial M/\partial y = \partial N/\partial x$

ملاحظة: الأمر odeadvisor في البرنامج الجاهز with(DEtools) يستخدم لبيان نوع المعادلة التفاضلية.

بذلك يكون البرنامج كالآتي:

```
> with(DEtools):

M:=(x,y)->cos(x)+2*x*exp(y):
N:=(x,y)->sin(y)+x^2*exp(y)-1:
```

```
diff(M(x,y),y):
diff(N(x,y),x):
eq:=M(x,y(x))+N(x,y(x))*diff(y(x),x)=0;
odeadvisor(eq);
sol:=dsolve(eq,y(x));
```

$$eq := \cos(x) + 2\,x\,e^{y(x)} + \left(\sin(y(x)) + x^2\,e^{y(x)} - 1\right)\left(\frac{d}{dx}\,y(x)\right) = 0$$

$$[\_exact]$$

$$sol := \sin(x) + x^2\,e^{y(x)} - \cos(y(x)) - y(x) + \_C1 = 0$$

لاحظ أن الأمر odeadvisor يشير إلى إن نوع المعادلة هو exact (تامة).

مثال (٨): حل المعادلة التفاضلية الآتية $(y^2+2xy)\,dx - x^2dy = 0$

يمكن كتابة المعادلة أعلاه بالصيغة الآتية $\dfrac{dy}{dx} = F\left(\dfrac{y}{x}\right)$

أي أن $dy/dx = (y^2+2xy)/x^2$

و هي معادلة متجانسة يمكن حلها بالفرضية $y = ux$

وبذلك يكون الحل كالآتي

$$\left(y^2+2xy\right)dx - x^2dy = 0$$

$$\left(u^2x^2+2ux^2\right)dx - x^2(u\,dx + x\,du) = 0$$

$$\left(u^2+2u\right)dx - (u\,dx + x\,du) = 0$$

$$\left(u^2+u\right)dx = -x\,du$$

$$\frac{1}{u\,(u+1)}du = -\frac{1}{x}dx.$$

وبالتكامل نحصل على

$$\ln|u| - \ln|u+1| = -\ln|x| + C$$

$$\frac{u}{u+1} = Cx$$

$$\frac{\frac{y}{x}}{\frac{y}{x}+1} = Cx$$

$$y = \frac{Cx^2}{1-Cx}.$$

عليه سيكون البرنامج بالشكل الآتي

```
> with(DEtools):
eq:=(y(x)^2+2*x*y(x))-x^2*diff(y(x),x)=0:
odeadvisor(eq);
solve((y/x)/(y/x+1)=c*x,y);
```

$$[[\_homogeneous, class\ A], \_rational, \_Bernoulli]$$

$$-\frac{c\,x^2}{-1+c\,x}$$

أو يمكن استخدام الأمر dsolve مباشرة وكالاتي

```
> eq:=(y(x)^2+2*x*y(x))-x^2*diff(y(x),x)=0:
  sol:=dsolve(eq,y(x));
```

$$sol := y(x) = \frac{x^2}{-x + \_C1}$$

```
> toplot:=seq(subs(_C1=i,rhs(sol)),i=-5..5):
plot([toplot],x=-5..5,view=[-5..5,-5..5],scaling=constrained,color=black);
```



مثال (٩): حل المعادلة التفاضلية الآتية عددياً $\frac{dy}{dt} = \left(t^2 - y^2\right)\sin y,\ y(0) = -1$

محدداً y(1) ثم ارسم $y(t),\ -1 \le t \le 10$

```
> sol:=dsolve({diff(y(t),t)=(t^2-y(t)^2)*sin(y(t)),y(0)=-1},y(t),numeric);
sol(1);
with(plots):    odeplot(sol,[t,y(t)],t=0..10,color=black);
```

$$sol := \mathbf{proc}(x\_rkf45)\ \ldots\ \mathbf{end\ proc}$$

$[t = 1., y(t) = -0.76601299$